
// pfc_save event, overrides ancestor 

 

INTEGER li_Return, i 

datetime ldt_today 

 

// Parent and child saves are dependent on each other, turn off AutoCommit 

// Control updates so that "All or None" are commited 

SQLCA.AutoCommit = false 

ldt_today = datetime(today(), now()) 

 

// Update other header columns before save 

dw_tariff_tier.AcceptText() 

if dw_tariff_tier.getitemstatus(1, 0, Primary!) = NewModified! or dw_tariff_tier.getitemstatus(1, 0, Primary!) = DataModified! then 

 dw_tariff_tier.object.ptt_cd_company[1] = gnv_app.ids_userprofile.Object.usr_cd_company[1] 

 dw_tariff_tier.object.ptt_id_user[1] = gnv_app.of_GetUserId() 

 dw_tariff_tier.object.ptt_dt_user[1] = ldt_today 

  

 if ib_active or is_mode = "copy" then 

  dw_tariff_tier.object.ptt_dt_active[1] = ldt_today 

  ib_active = false 

 end if 

end if 

 

li_Return = dw_tariff_tier.update(false, false)  // No AcceptText, No reset update flags 

IF li_Return <> 1 THEN  

 ROLLBACK USING SQLCA; 

 SQLCA.AutoCommit = true 

 MessageBox("Save Failed", "Tiered rate header save failed. Please contact the System Administrator", Exclamation!, Ok!) 

 RETURN -1 

END IF 

 

// Get the tier id and update tiers 

il_id_tier = dw_tariff_tier.object.ptt_id_tier[1] 

MessageBox("Save", "il_id_tier: " + string(il_id_tier), Information!, Ok!) 

dw_tariff_tier_tier.AcceptText() 



for i = 1 to dw_tariff_tier_tier.rowcount() 

 if dw_tariff_tier_tier.getitemstatus(i, 0, Primary!) = NewModified! or dw_tariff_tier_tier.getitemstatus(i, 0, Primary!) = DataModified! then 

  dw_tariff_tier_tier.object.ttt_id_tier[i] = il_id_tier 

  dw_tariff_tier_tier.object.ttt_id_user[i] = gnv_app.of_GetUserId() 

  dw_tariff_tier_tier.object.ttt_dt_user[i] = ldt_today 

 end if 

next 

 

li_Return = dw_tariff_tier_tier.update(false, false)  // No AcceptText, No reset update flags 

IF li_Return <> 1 THEN  

 ROLLBACK USING SQLCA; 

 SQLCA.AutoCommit = true 

 MessageBox("Save Failed", "Tiered rate tiers save failed. Please contact the System Administrator", Exclamation!, Ok!) 

 RETURN -1 

END IF 

 

// Commit both DW's 

COMMIT USING SQLCA; 

 

// Reset update flags for DW's 

dw_tariff_tier.resetupdate() 

dw_tariff_tier_tier.resetupdate() 

 

// Turn auto-commit back on 

SQLCA.AutoCommit = true 

 

messagebox("Save","Save Successful") 

 

return 1 


